
Appendix 
L-COSY pulse program 

This is the L-COSY pulse program implemented on the Bruker BioSpin 9.4 and 17.6T 
MR spectrometers used for the research presented in this thesis. 
  
; lcosy.ppg - 
 
#include <Avance.incl> 
#include <DBX.include> 
 
#include <PrepModulesHead.mod> 
 
define delay de4 
"de4=0.00002-de+depa" 
 
define delay decodur 
"decodur=0.00002-de+depa" 
#include <decoacq.mod> 
define loopcounter lds={$NDummyScans} 
 
preset off 
 
#include <MEDSPEC.include> 
 
;-------D/P-----GRAD----------------------------RF/ACQ------------ 
start, 4u      
av, d1    fq8b:f1 
;---------Triggered data ----------------------------------------- 
#include <TriggerPerSlice.mod> 
   
;---------noe  module----------------------------------------- 
#include "noe.mod" 
;---------ws & ovs module----------------------------------------- 
#include "WsOvs.mod" 
 
;---------localisation  module------------------------------------ 
 1u ; reset:f1 
 d4 grad{  (t1) | (0.0) | (0.0) }  fq1:f1 
 d6     gatepulse 1 
 (p1:sp1 ph1):f1 
 d6 
 d7 groff    
 goto 10 
nav, 1u 
 if ( PVM_ppgFlag4 == Yes) {    ; Retro frequency lock 

 d4 grad{  (t1) | (0.0) | (0.0) }  fq1:f1 
 d6     gatepulse 1 
 (p1:sp0 ph1):f1 
 d6 
 d7 groff  
 } 
10      d8 grad{  (t4) |  (t6) |  (t7) }  
 d4 grad{ (0.0) |  (t2) | (0.0) }   fq1:f1 
 d6     gatepulse 1 
 (p2:sp2 ph2):f1 
 d6 
 d8 grad{  (t5) |  (t6) |  (t7) } 
 d9 groff 
 
        d0  ; COSY time evolution 
   
 ;pulse adjustments: 
 if ( PVM_ppgFlag3 == Yes ) { 
 d12 grad{ (t11) | (t12) | (t13) } 
 d4 groff  
 } 
 d10 grad{  (t8) |  (t9) | (t10) } fq1:f1 
 d4 grad{ (0.0) | (0.0) |  (t3) } 
 d6     gatepulse 1 
 (p3:sp3 ph3):f1 
 d6 
 d10 grad{  (t8) |  (t9) | (t10) } 
 d11 groff     
 
;---------data acquisition--------------------------------------------  
 ;pulse adjustments: 
 if ( PVM_ppgFlag3 == Yes )  
 { 
          d4 grad{ (t14) | (t15) | (t16) }  
   de4     REC_ENABLE 
                ADC_INIT_B(ph31, ph30) 
   aqq     ADC_START  
 } 
 else  
 { 
subr decoup(decodur,ph31,ph30) 
        } 
        3m groff 
 if(OPT_RFLOnOff == On) 
        { 
     1m reload B0 
 } 
 else 
 { 
     1m 
 } 
 1m     ADC_END  

;---------navigator loop --------------------------------------------- 
 lo to nav times l1 
 
;---------dummy scans ------------------------------------------------ 
 "lds = lds - 1"   
 if "lds>=0" goto start  
 
;---------averaging & phase cycling----------------------------------- 
 if ( ACQ_scan_type == Scan_Experiment ) { 
 1u     ipp1 
 1u     ipp2 
 1u      ipp3 
 1u      ipp31 
 } 
 lo to start times NA  
 1u     rpp1 
 1u     rpp2 
 1u      rpp3 
 1u      rpp31 
 
        if(ACQ_scan_type != Setup_Experiment) 
        { 
          1u id0 
          1u id1 
        } 
        lo to start times ACQ_size[1] 
       
        1u          
        lo to av times NR 
 SETUP_GOTO(av) 
exit 
 
ph0 = 0 
ph1 = 0 0 0 0 0 0 0 0  0 0 0 0 0 0 0 0 
ph2 = 0 0 1 1 0 0 1 1  2 2 3 3 2 2 3 3 
ph3 = 0 2 0 2 1 3 1 3  0 2 0 2 1 3 1 3 
ph30 = 0 
ph31= 0 0 2 2 2 2 0 0  0 0 2 2 2 2 0 0 

